
بسم الله السميع العليم 

سلسلة تقنيات تحسين واجهات البرامج 

النوافذ الطبقية 

 الطيب رغيد:  بقلم

 

إلا فئة معينة ... وهو امر حقيقي مسلم به في عيون كل الناس العادين ! .... قيل قديما بأن إرضاء الناس غاية لا تدرك 
 ..و هم المبرمجون !! ... من الناس 

وذلك لكون الإنسان العادي يمكنه ان !! ... غاية لا تدرك ) خدمالمست(إرضاء ..... لقالوا ... الوحيدون الذين إذا سألتهم  مه
 !ويستريح "صحيح إرضاء الناس غاية لا تدرك" :ببساطة  يقول

 !!فلن يبقى هناك من يشتري برامجه ... لإنه إن فعل ... ان يقول هذا  هغير ان المبرمج لا يمكن

للمبرمج الإعتماد الى الكثير من الحيل و الخدع و  يمكن.... اقول ... على العموم وبعيدا عن احزاني و شجوني البرمجية 
لأن المستخدم ... طبعا المستخدم العادي ) ... العادي(المستخدم و إستحسان  إنتباه لجذباللعب بالاكواد البرمجية 

و هل حقق وسهولة الاستخدام و لكنه سيلتفت الى المضمون ... ألوانك  الخبير لن يلتفت الى جمال واجهاتك و تناسق
 .غرضه الرئيسي ام لا كجبرنام

تقنيات تحسن (التي اسمتيها ... و في هذه السلسلة ... الشكل و المضمون .... فلا بأس ان نعتني بالاثنتين  عليه و
اتمنى ان لا ) ... ما اقبح الكلمات و أنبل الغرض ( عالى بعض الحيل و الخدع سنناقش سويا بإذن الله ت  )واجهات البرامج

....  ناحبصا دقو اهنم يهتاو نيلاتحم  و نيعداخم لہاب ذايعلال تفهم اننا في هذه السلسة س

 

 .ا العزيزنمل في عيون مستخدمدوات لجعلها تبدوا اجو النوافذ و الأالوينذوز إن كنا سنخدع فإننا سنخدع 

 

الخدعة الأولى 

النوافذ الطبقية 

 

 عادية يمكن التعديل عليها ساعة إنشائها و جعلها تتصف نافذةو هي  Form1كا الفورم العادية النافذة ببساطة هي 
... وهكذا ... لها عنوان ام لا , ثابته, متحركة, افةمعينة مثل شفبصفات 

ايضا عن التابعة للفورم ويمكن ذلك   CreateParamsولكي خواص النوافذ توجد عدة طرق اما بإستخدام الخاصية 
 SetWindowLongكالدالة  APIدوال الـ طريق 

هي ان نجعل الفورم عبارة عن فورم  و WS_LayeredWindowو عن طريق اي الطرق سنقوم بتغيير خواص الفورم الى 
: كالتالي ... طبقية 

Dim exStyle As Int32 = GetWindowLong(Me.Handle.ToInt32,GWL_EXSTYLE) 
 
SetWindowLong(Me.Handle.ToInt32,GWL_EXSTYLE, exStyle Or WS_EX_LAYERED) 
 



...   GetWindowLongو  API SetWindowLongمن الكود السابق ستعرف اني قمت بإستخدام دوال الـ  

 

! ا اجعل النافذة طبقية ؟هو لماذ... طبقية و لكن السؤال الحقيق من الجميل ان نكون قد تعرفنا على كيفية جعل النافذة 
ى شكل صورة لفية الفورم شفافة او علتغيير خصائصها مثل جعل خجعل النافذة طبقية يمكنك .. 

 

 Regionsاو العمل مع الـ   TransparentColorالخاصية ... طبعا توجد الكثير من الطرق لجعل الفورم تدعم الشفافية مثل 
. التابعة للفورم Opacityكما يمكن ايضا ان نحصل على نوع من الشفافية عن طريق الـخاصية ... وما الى ذلك 

 

الداعم للشفافية و لنقل هذه  PNGلنفترض ان لدينا صورة جميلة من النوع .... غير ان ما اتكلم عنه هو شيء مختلف 
: الصورة هي صورة النحلة 

 

 

 

 

 

و لنفرض انني رغبت في ان اجعل الفورم ... النحلة كما تروق لي بالفعل  مقارب لذوقي فسوف تروق لك اذا كان ذوقك
: الشكل التالي ... لتبدو مقاربة الى .... تأخذ شكل هذه النحلة 

  



 BackgroundImageفكرنا بعمل فورم وذلك بجعل الصورة خلفية لها في الخاصية فلربما ... اذا اردنا ان نقوم بذلك برمجيا 
الاجزاء البيضاء وبهذا تصبح .. ابيض ايضا  transparentColorابيض مثلا و نجعل الخاصية  BackColorلفية نجعل لون الخو 

: شفافة كالتالي 

 

 

 

 

: و الناتج عن ذلك شيئا مشابا للتالي 

 

 

 

 

 

 

 

    Private Sub Form1_Load(……) Handles MyBase.Load 
        Me.BackColor = Color.White 
        Me.TransparencyKey = Color.White 
    End Sub 



جميل غير انك اذا أمعنت النظر ستجد ان الاطراف  !! .... ود ن فقط من الككون قد حقننا شفافية جميلة و بسطريو بهذا ن
اما اذا رجت الى الصورة التي طلبناها .... غير ظاهرة جزئيا ) النحلة(كما ان الاجزاء تحت الصورة ... يوجد عندها لو أبيض 

: فهي كالتالي 

 

ستلاحظ ان 
اطراف الصورة 

و لا نظيفة كليا 
يوجد اثر للون 

الابيض كما ان 
اجنة النحلة 
في الصورة 

شفافة جزئيا  
يمكنك رؤية ما 

. تحتها

 

 

: التابعة للفورم فستجد ان الصورة كلها اصبحت شفافة كالتالي  Opacityاذا اخترت ان تقوم بتعديل الخاصية و

 

 

 

 

 

 

 

 

 

 

لا تهتم  Opacityستجد ان الصورة كلها اصبحت شفافة حتى الاجزاء التي ينبغي أن لا تكون شفافة وذلك لأن الخاصة 
. بشكل عام مبتفاصيل الصورة و إنما تهتم بشفافية الفور

 

 Alphaو ثم تكون لدينا صورة ذات تدعم الشفافية  Layered Windowان تكون اولا الفورم طبقية و هنا نحتاج الى 
Blending  و تجد هذا غالبا في الصور من النوع (وفيها أجزاء شفافة و أخرى شبه شفافة و اخرى معتمةPNG   (

 



لجعل الفورم شفافة وبحسب  APIsة تجعل بالامكان إستخدام بعضا من دوال الـ لهذا الغرض كله قمت بتصميم فئة بسيط
.... الصورة التي ترديها 

للتحكم  APIوهي عبارة عن فئة بسيطة تقوم بإستخدام بعض دوال الـ  rtaLayeredFormاسم الفئة التي صممتها هو 
ستحتاج الى كتابة شيء مثل هذا في ... لكي تستخدم الفئة وتجعل الفورم بشكل اي صورة تريدها ... بمظهر الفورم 

: للفورم  Loadالحدث 

 

 

 

 

 

: امثلة على اشكال الفورم التي يمكن ان تنشئها 

.... عبارة عن فورمات في برنامجنا منتشرة هنا و هناك الاعلى هنا إن ما تراه في 

 

  

    Private Sub Form1_Load(……) Handles MyBase.Load 
 
        Dim Layer As New rtaLayeredForm() 
        Layer.SetImage(Me, Me.BackgroundImage) 

    End Sub 



Public Class rtaLayeredForm 
 
    Private mParent As Form 
    Public ReadOnly Property Parent() As Form 
        Get 
            Return mParent 
        End Get 
    End Property 
 
    Private mMovable As Boolean = True 
    Public Property Movable() As String 
        Get 
            Return mMovable 
        End Get 
        Set(ByVal value As String) 
            mMovable = value 
        End Set 
    End Property 
 
    Public Sub SetImage(ByVal frm As Form, ByVal btm As Bitmap) 
 
        Try 
            mParent = frm 
 
            AddHandler frm.MouseDown, AddressOf OnMouseDown 
            AddHandler frm.MouseMove, AddressOf OnMouseMove 
 
            Dim exStyle As Int32 = APIs.GetWindowLong(frm.Handle.ToInt32, APIs.GWL_EXSTYLE) 
 
            APIs.SetWindowLong(frm.Handle.ToInt32, APIs.GWL_EXSTYLE, exStyle Or 
APIs.WS_EX_LAYERED) 
            If Not (Bitmap.IsCanonicalPixelFormat(btm.PixelFormat)) Or Not 
(Bitmap.IsAlphaPixelFormat(btm.PixelFormat)) Then 
                Throw New ApplicationException("The picture must be 32bit picture with alpha 
channel.") 
            End If 
 
            Dim hDC As Int32 = APIs.GetDC(0) 
            Dim hBtm As Int32 
            Dim memDC As Int32 = APIs.CreateCompatibleDC(hDC) 
            Dim Bits As Int32 
 
            Dim dstPos As New APIs.POINTAPI(frm.Left, frm.Top) 
            Dim srcPos As New APIs.POINTAPI(0, 0) 
            Dim Siz As New APIs.SizeAPI(btm.Width, btm.Height) 
            Dim Opt As New APIs.BLEND_OPTIONS 
 
            Opt.SourceConstantAlpha = 255 
            Opt.AlphaFormat = 1 
            Opt.BlendOp = 0 
            Opt.BlendFlags = 0 
 
            Try 
                hBtm = btm.GetHbitmap(Color.Black).ToInt32 
                Bits = APIs.SelectObject(memDC, hBtm) 
 
                APIs.UpdateLayeredWindow(frm.Handle, hDC, dstPos, Siz, memDC, srcPos, 0, Opt, 2) 
            Catch ex As Exception 
 
            End Try 
 
            If hBtm <> 0 Then 
                APIs.SelectObject(memDC, Bits) 
                APIs.DeleteObject(hBtm) 
            End If 
 
            APIs.ReleaseDC(0, hDC) 



            APIs.DeleteDC(memDC) 
 
        Catch ex As Exception 
 
        End Try 
    End Sub 
 
 
    Private mLastLocation As Point 
    Private Sub OnMouseDown(ByVal sender As Object, ByVal e As 
System.Windows.Forms.MouseEventArgs) 
        mLastLocation = e.Location 
    End Sub 
 
    Private Sub OnMouseMove(ByVal sender As Object, ByVal e As 
System.Windows.Forms.MouseEventArgs) 
        If (mMovable) AndAlso (e.Button = MouseButtons.Left) Then 
            Dim pos As Point = Parent.PointToScreen(e.Location) 
 
            pos.Offset(-mLastLocation.X, -mLastLocation.Y) 
            Parent.Location = pos 
        End If 
    End Sub 

End Class 

 

 

 :وهي كالتالي ... التي استخدمتها  APIsالـ  وهذه الفئة بكل

Private Class APIs 
 
        Public Declare Function CreateCompatibleDC Lib "gdi32.dll" (ByVal hdc As Int32) As Int32 
        Public Declare Function GetDC Lib "user32.dll" (ByVal hwnd As Int32) As Int32 
        Public Declare Function SelectObject Lib "gdi32.dll" (ByVal hdc As Int32, ByVal hObject 
As Int32) As Int32 
        Public Declare Function ReleaseDC Lib "user32.dll" (ByVal hwnd As Int32, ByVal hdc As 
Int32) As Int32 
        Public Declare Function DeleteDC Lib "gdi32.dll" (ByVal hdc As Int32) As Int32 
        Public Declare Function DeleteObject Lib "gdi32.dll" (ByVal hObject As Int32) As Int32 
        Public Declare Function UpdateLayeredWindow Lib "user32.dll" (ByVal hWnd As Int32, ByVal 
hdcDst As Int32, ByRef pptDst As POINTAPI, ByRef psize As SizeAPI, ByVal hdcSrc As Int32, ByRef 
pptSrc As POINTAPI, ByVal crKey As Int32, ByRef pblend As BLEND_OPTIONS, ByVal dwFlags As Int32) 
As Int32 
        Public Declare Function GetWindowLong Lib "user32.dll" Alias "GetWindowLongW" (ByVal hwnd 
As Int32, ByVal nIndex As Int32) As Int32 
        Public Declare Function SetWindowLong Lib "user32.dll" Alias "SetWindowLongW" (ByVal hwnd 
As Int32, ByVal nIndex As Int32, ByVal dwNewLong As Int32) As Int32 
 
        Public Const GWL_EXSTYLE As Int32 = -20 
        Public Const WS_EX_LAYERED As Int32 = &H80000 
 
        <StructLayout(LayoutKind.Sequential, Pack:=1)> _ 
        Public Structure BLEND_OPTIONS 
            Public BlendOp As Byte 
            Public BlendFlags As Byte 
            Public SourceConstantAlpha As Byte 
            Public AlphaFormat As Byte 
        End Structure 
 
        <StructLayout(LayoutKind.Sequential)> _ 
        Public Structure SizeAPI 
            Public cx As Int32 
            Public cy As Int32 



 
            Sub New(ByVal cx As Int32, ByVal cy As Int32) 
                Me.cx = cx 
                Me.cy = cy 
            End Sub 
        End Structure 
 
        <StructLayout(LayoutKind.Sequential)> _ 
        Public Structure POINTAPI 
            Public x As Int32 
            Public y As Int32 
 
            Sub New(ByVal x As Int32, ByVal y As Int32) 
                Me.x = x 
                Me.y = y 
            End Sub 
        End Structure 
 
        Public Declare Function ExtCreateRegion Lib "gdi32.dll" ( _ 
          ByRef lpXform As XFORM, _ 
          ByVal nCount As Int32, _ 
          ByRef lpRgnData As RGNDATA) As Int32 
        <StructLayout(LayoutKind.Sequential)> _ 
        Public Structure XFORM 
            Public eM11 As Single 
            Public eM12 As Single 
            Public eM21 As Single 
            Public eM22 As Single 
            Public eDx As Single 
            Public eDy As Single 
        End Structure 
 
        <StructLayout(LayoutKind.Sequential)> _ 
        Public Structure RGNDATA 
            Public rdh As RGNDATAHEADER 
            Public Buffer As Byte 
        End Structure 
 
        <StructLayout(LayoutKind.Sequential)> _ 
        Public Structure RGNDATAHEADER 
            Public dwSize As Int32 
            Public iType As Int32 
            Public nCount As Int32 
            Public nRgnSize As Int32 
            Public rcBound As RECT 
        End Structure 
 
        <StructLayout(LayoutKind.Sequential)> _ 
        Public Structure RECT 
            Public Left As Int32 
            Public Top As Int32 
            Public Right As Int32 
            Public Bottom As Int32 
        End Structure 
 
End Class 

 

 



 


